
Mathematica

Grafika

Polecenie Graphics3D[{ {dir1, prim1}, {dir2, prim1 }, ...}, opcje ]
rysuje w przestrzeni 3 wymiarowej obiekty graficzne prim1, prim1, ... w
sposób wskazany przez dyrektywy dir1, dir2, ....

Zadanie 1. Narysować punkty o współrzędnych (−1,−2, 1), (2, 3, 1) i połą-
czyć je odcinkiem.
Rozwiązanie:

p1 = {-1, -2, 1}; p2 = {2, 3, 1};

Graphics3D[
{ {PointSize [0.03], Point[p1], Point[p2]}, Line[{p1, p2}]},
Axes -> True ]

Zadanie 2. Narysować (kolorem czerwonym) sferę o promieniu 1 i o środku
w punkcie (−1,−2, 1).
Rozwiązanie:
Polecenie Sphere[{x, y, z}, r] rysuje sferę o promieniu r i środku (x, y, z).

Graphics3D[ {Red, Sphere[p1, 1]},
Axes -> True]

Zadanie 3. Narysować sferę o promieniu 1 i o środku w punkcie (0, 0, 0).
Narysować punkt w środku sfery i punkt na sferze. Połączyć je odcinkiem.
Sfera ma być półprzeźroczysta.
Rozwiązanie:

p1 = {0, 0, 0}; p2 = {0, 1, 0};

Graphics3D[{{Opacity[.6], Red, Sphere[p1, 1]}, {Black,
PointSize[0.03], Point[p1], Point[p2]}, Thick, Line[{p1, p2}]},
Axes -> True]

Zadanie 4. Narysować 200 sfer o losowych środkach, promieniach i kolorach.
Wsk. Użyć funkcji RandomReal[x], której wartością jest losowa liczba z prze-
działu [0, x]. Na przykład RandomReal[0.1] wygeneruje losową liczbę z prze-
działu [0, 0.1], a RandomReal[1, {3}] jest losową trójką liczb leżących w
przedziale [0, 1].
Rozwiązanie:

Graphics3D[
Table[{Hue[RandomReal[]],
Sphere[RandomReal[1, {3}], RandomReal[0.1]]}, {200}]]

Wyjaśnienie: Hue[h] jest kodem koloru o numerze h. Table[w, {n}] tworzy
listę n elementów w.

Zadanie 5. Polecenie Cuboid[{x0, y0, z0}] rysuje sześcian jednostkowy
o krawędziach równoległych do osi współrzędnych. Współrzędne jego wierz-
chołka położonego najdalej w kierunkach przeciwnych do zwrotów osi to
(x0, y0, z0).

Cuboid[{x0, y0, z0}, {x1, y1, z1}] rysuje prostopadłościan o krawę-
dziach równoległych do osi współrzędnych. Współrzędne jego wierzchołka
położonego najdalej w kierunkach przeciwnych do zwrotów osi to (x0, y0, z0),
a współrzędne wierzchołka przeciwnego to (x1, y1, z1).
Wydaj polecenia
Graphics3D[Cuboid[{0, 0, 0}]] oraz

Graphics3D[{Yellow, Cuboid[{0, 0, 0}, {1, 3, 1}],
Blue, Cuboid[{2, 1, 1}, {4, 2, 3}]}]

Zadanie 6. Sposób narysowania krawędzi prostopadłościanu określa dyrek-
tywa EdgeForm[g], a jego ścian - dyrektywa FaceForm[g].
Wydaj polecenia

Graphics3D[{EdgeForm[{Thick, Blue}], FaceForm[{Pink, Opacity[0.8]}],
Cuboid[]}, Boxed -> False]

Graphics3D[
Table[{EdgeForm[Opacity[.3]], Hue[RandomReal[]],
Cuboid[RandomReal[4, 3]]}, {40}]]

Wyjaśnienie: RandomReal[4, 3] tworzy listę trzech liczb losowych z odcinka
[0, 4).

Zadanie 7. Polecenie Cylinder[{{x1, y1, z1}, {x2, y2, z2}},r] ry-
suje walec o promieniu r którego osią symetrii jest odcinek od punktu (x1, y1, z1)
do punktu (x2, y2, z2).
Wydaj polecenia
Graphics3D[Cylinder[{{0, 0, 0}, {1, 1, 1}}, 1/2]]

Graphics3D[
Table[{EdgeForm[Opacity[.3]], Hue[RandomReal[]],
Cylinder[RandomReal[10, {2, 3}], 0.2]}, {20}]]

Graphics3D[{Opacity[.3], EdgeForm[Opacity[.3]],
Table[Cylinder[{{0, 0, 0}, {0, 0, 2 r}}, r], {r, 1, 5}]},
Boxed -> False]

Zadanie 8. Polecenie Cone[{{x1, y1, z1}, {x2, y2, z2}},r] rysuje sto-
żek promieniu podstawy r, środku podstawy (x1, y1, z1) i wierzchołku (x2, y2, z2).
Wydaj polecenia
Graphics3D[Cone[{{0, 0, 0}, {1, 1, 1}}, 1/2]]

Graphics3D[
Table[{EdgeForm[Opacity[.3]], Hue[RandomReal[]],
Cylinder[RandomReal[10, {2, 3}], 0.2]}, {20}]]

Graphics3D[{Opacity[.3], EdgeForm[Opacity[.3]],
Table[Cone[{{0, 0, 0}, {0, 0, 2 r}}, r], {r, 1, 5}]},
Boxed -> False]

Zadanie 9. Polecenie Polygon[p1, p2, ...] rysuje wypełniony wielokąt,
o wierzchołkach p1, p2, . . .
Wydaj polecenia
Graphics3D[Polygon[{{1, 0, 0}, {1, 1, 1}, {0, 0, 1}}]]

Zadanie 10. Narysuj sześcian A,B,C,D,A1, B1, C1, D1 i jego przekrój płasz-
czyzną przechodzącą przez wierzchołki A,B,C1, D1.
Utwórz sześcian:

szescian = {EdgeForm[{Thick, Blue}], FaceForm[{Pink, Opacity[0.1]}],
Cuboid[]}

Utwórz przekrój:

pl = Polygon[ {{0, 0, 0}, {1, 0, 0}, {1, 1, 1}, {0, 1, 1}}];

i wreszcie grafika:
Graphics3D[{szescian, {Red, pl}}, Boxed -> False]

Zadanie 11. Polecenie Rotate[g, u, v] obraca obiekt g o kąt u wokół
wektora v.
Następujące polecenie dynamicznie obraca sześcian o kąt = 2πk

20
, k = 1, . . . 20.

Osią obrotu jest prosta przechodząca przez punkty (0, 0, 0) i (1, 1, 1).

pr = {-1, 2};

Manipulate[
Graphics3D[ {Rotate[Cuboid[], 2 Pi k/20, {1, 1, 1}], {Red, Thick,
Line[{{-1, -1, -1}, {2, 2, 2}}]}}
, PlotRange -> {pr, pr, pr}], {k, 0, 20}]

I jeszcze to samo, tylko statycznie:

pr = {-1, 2};
t = Table[Rotate[Cuboid[], 2 Pi k/20, {1, 1, 1}], {k, 30}];
Graphics3D[t]

Zadanie 12. Narysować 5 brył platońskich.
Polecenie:
PolyhedronData["Cube"]
W miejsce "Cube" wpisać po kolei
"Dodecahedron", "Icosahedron", "Octahedron", "Tetrahedron"

